

# Solution du TP sur les fractions continues

O. Marguin - 11 mai 2011

## Exercice 1

```
> restart;
> with(numtheory):
> seq(cfrac(sqrt(k^2+1), 'periodic', 'quotients'), k=1..10);
[[1], [2]], [[2], [4]], [[3], [6]], [[4], [8]], [[5], [10]], [[6], [12]], [[7], [14]], [[8],
[16]], [[9], [18]], [[10], [20]]
```

Il semble que, pour tout entier  $n > 0$ , le développement en fraction continue de  $\sqrt{n^2 + 1}$  soit égal à  $(n, 2n, 2n, 2n, \dots)$ .

Pour le démontrer, il suffit de vérifier que l'on a  $\sqrt{n^2 + 1} = n + \frac{1}{x}$  où  $x$  est solution de

l'équation  $x = 2n + \frac{1}{x}$  :

```
> s:=solve(x=2*n+1/x, x);
s:=n+sqrt(n^2+1), n-sqrt(n^2+1)
```

On ne retient que la solution positive  $s[1]$ .

```
> simplify(sqrt(n^2+1)-n-1/s[1]);
0
```

CQFD !

## Exercice 2 : approximants de Padé

```
> restart;
> Pade:=proc(expr, x, k)
local S, a, n, z;
S:=convert(series(expr, x=0, 20), polynomial);
a:=coeff(S, x, 0);
n:=ldegree(S-a, x);
if k=0 then return a
else
z:=convert(series(x^n/(expr-a), x=0, 20), polynomial);
return a+x^n/(Pade(z, x, k-1))
end if
end proc;
> P:=Pade(tan(x), x, 4); normal(P);
```


$$P := \frac{x}{1 + \frac{x^2}{-3 + \frac{x^2}{5 - \frac{x^2}{7}}}}$$

$$= \frac{5x(-21 + 2x^2)}{105 - 45x^2 + x^4}$$

```
> s:=seq(Pade(tan(x), x, k), k=0..5);
```

$$s := 0, x, \frac{x}{1 - \frac{x^2}{3}}, \frac{x}{1 + \frac{x^2}{-3 + \frac{x^2}{5}}}, \frac{x}{1 + \frac{x^2}{-3 + \frac{x^2}{5 - \frac{x^2}{7}}}}, \frac{x}{1 + \frac{x^2}{-3 + \frac{x^2}{5 + \frac{x^2}{-7 + \frac{x^2}{9}}}}}$$

```
> plot([tan(x), s[4]], x=-10..10, color=[black, red], thickness=[1, 2], view=[-4.6..4.6, -10..10]);
```



On obtient une bonne approximation de la fonction tangente, même au voisinage des premiers pôles.